Supporting Information for:
Improving the Analysis of Phase-Separated Bio-Fuel Samples with Slice-Selective Total Correlation NMR Spectroscopy

Jaskamal Singh Khangura[†], Bridget Tang[†], Katie Chong[‡] and Robert Evans[†]*,
† Chemical Engineering and Applied Chemistry, College of Engineering and Physical Sciences, Aston University, Birmingham, B4 7ET, UK, United Kingdom.
‡ Energy and Bioproducts Research Institute, Aston University, Aston Triangle, Birmingham B4 7ET, United Kingdom.

Contents

SI.1 Pulse sequence schematic and details

Schematic of NMR Experiment



Figure S1. 1D ^{1}H slice-selective NMR pulse sequence used and schematic of excitation (not drawn to scale).

Corresponding pulse sequences for Bruker spectrometer.

```
#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"acqt0=0"
"d11=p11-10u"

  1 ze
  2 d1
    10u pl0:f1
    10u gron1
      d11
    10u
    10u groff
    4u pl1:f1
    10u gron0
    p11:sp1:f1 ph1:r
    10u groff
    4u
    go=2 ph31
    30m mc #0 to 2 F0(zd)
  exit

      ph1=0 2 2 0 1 3 3 1
      ph31=0 2 2 0 1 3 3 1
```

Pulse sequence originally published in *Energy Fuels* 2017, 31, 4135–4142.

```
;Slice selective 2D TOCSY
;use G4 selective pulse for single slice/ REBURP inversion for double slice selection
;
;based on ...
;
;mlevph
;avance-version (07/04/04)
;homonuclear Hartman-Hahn transfer using MLEV17 sequence
;   for mixing
;using two power levels for excitation and spinlock
;phase sensitive
;
;A. Bax & D.G. Davis, J. Magn. Reson. 65, 355-360 (1985)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Delay.incl>

"acqt0=0"
"p5=p6*.667"
"p7=p6*2"
"d11=p11-10u"
"in0=inf1"
"d0=in0/2-p1*2/3.1416-4u"

"SCALEF=p7*2/p5+0.5"
"FACTOR1=((d9-p17*2)/(p6*64+p5))/SCALEF+0.5"
"l1=FACTOR1*SCALEF"

1 ze
2 d1
3 20u pl0:f1
  10u gron1
       d11
  10u
  10u groff
  4u pl1:f1
  10u gron0
  p11:sp1:f1 ph1:r
  10u groff
  4u
  d0
```

```
  4u pl10:f1
  (p17 ph26)
                                                ;begin MLEV17
4 (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph22 p7 ph23 p6 ph22)
  (p6 ph24 p7 ph25 p6 ph24)
  (p6 ph24 p7 ph25 p6 ph24)
  (p5 ph23)
  lo to 4 times l1
                                                ;end MLEV17
  (p17 ph26)
  go=2 ph31
  d1 mc #0 to 2 F1PH(ip1, id0)
exit

ph1=0 2 2 0 1 3 3 1
ph22=3 1 3 1 0 2 0 2
ph23=0 2 0 2 1 3 1 3
ph24=1 3 1 3 2 0 2 0
ph25=2 0 2 0 3 1 3 1
ph26=0 2 0 2 1 3 1 3
ph31=0 2 2 0 1 3 3 1

;pl1 : f1 channel - power level for pulse (default)
;pl10: f1 channel - power level for TOCSY-spinlock
;p1 : f1 channel -  90 degree high power pulse
;p5 : f1 channel -  60 degree low power pulse
;p6 : f1 channel -  90 degree low power pulse
;p7 : f1 channel - 180 degree low power pulse
;p11 : f1 channel - selective pulse for slice select
;p17: f1 channel - trim pulse                    [2.5 msec]
;d0 : incremented delay (2D)
;d1 : relaxation delay; 1-5 * T1
;d9 : TOCSY mixing time
;l1: loop for MLEV cycle: (((p6*64) + p5) * l1) + (p17*2) = mixing time
;inf1: 1/SW = 2 * DW
;in0: 1/(1 * SW) = 2 * DW
;nd0: 1
```

;NS: 8 * n
;DS: 16
;td1: number of experiments
;FnMODE: States-TPPI, TPPI, States or QSEQ

;Processing

;PHC0(F1): 180
;PHC1(F1): -180
;FCOR(F1): 1

Based on double-slice selective pulse sequence published in *Physical Chemistry Chemical Physics* 2022, 24, 17961-17965.

SI.2 Additional spectra of unseparated samples



Figure S2. 2D ^{1}H TOCSY spectra of three-component unseparated sample D.



Figure S3. 2D ^{1}H TOCSY spectra of three-component unseparated sample E.



Figure S4. 2D ^{1}H TOCSY spectra of three-component unseparated sample F.

SI.3 Additional spectra of separated samples



Figure S5. Slice-selective 2D ^{1}H TOCSY spectrum of upper layer of three-component separated sample C between 2 and 8 ppm (region indicated by box in Figure 3).



Figure S6. Slice-selective 2D ^{1}H TOCSY spectrum of lower layer of three-component separated sample C between 2 and 8 ppm (region indicated by box in Figure 3).